TRex 3.24

Use a computer to calculate Planck's radiation law for a temperature of 3000K, which is the temperature of a typical tungsten filament in an incandescent lightbulb. Plot the intensity versus wavelength.

a) How much of the power is in the visible region (400 - 700 nm) compared with the UV and the IR?

b) What is the ratio of the intensity at 400 nm and 700 nm to the wavelength with the maximum intensity?

Break Planck's law into constituents to create the function.

In[1]:=
```
constfactor = 2 * π * spdlt^2 * planck;
lambdafactor[λ_] := λ^5 * (Exp[(planck * spdlt)/(λ * boltzmn * T)] - 1)
fluxdensity[λ_] := constfactor/lambdafactor[λ]
fluxdensity[λ]
```

Out[4]= $\frac{2\pi\,\text{planck spdlt}^2}{\left(-1+e^{\frac{\text{planck spdlt}}{\text{boltzmn T}\,\lambda}}\right)\lambda^5}$

Tell *Mathematica* the values of the constants and write the function out.

In[5]:=
```
T = 3000;
planck = 6.63 * 10^-34;
spdlt = 3 * 10^8;
boltzmn = 1.38 * 10^-23;
vislow = 400 * 10^-9;
vishigh = 700 * 10^-9;
fluxdensity[λ]
```

Out[11]= $\frac{3.74918\times10^{-16}}{\left(-1+e^{4.80435\times10^{-6}/\lambda}\right)\lambda^5}$

Evaluate the function at the ends of the visible spectrum and the maximum.

In[12]:=
```
fluxdensity[λ]
fvlow = fluxdensity[λ] /. λ → vislow
fvhigh = fluxdensity[λ] /. λ → vishigh
FindMaximum[fluxdensity[λ], {λ, 0.95 * 10^-6}]
```

Out[12]= $\frac{3.74918\times10^{-16}}{\left(-1+e^{4.80435\times10^{-6}/\lambda}\right)\lambda^5}$

Out[13]= 2.22528×10^{11}

Out[14]= 2.33444×10^{12}

Out[15]= $\{3.10548\times10^{12},\ \{\lambda\to9.67621\times10^{-7}\}\}$

Name the peak of the emission as Imax for maximum intensity.

In[16]:=
```
Imax = 3.10548 * 10^12;
```

Thus the peak wavelenth of the emission of the tungstun filament is 968 nm which is in the IR band. Plot these with gridlines and ticks.

In[17]:=
```
PPlanck = Plot[fluxdensity[λ], {λ, 100 * 10^-9, 550 * 10^-8},
  BaseStyle → {FontColor → RGBColor[0, 0.5, 0], FontSize → 14},
  PlotStyle → {Brown}, AxesLabel → {"Wavelength μm", "Intensity W/m^2/λ"},
  GridLines -> {{vislow, vishigh, 0.96765 * 10^-6}, {fvlow, fvhigh, 3.105 * 10^12}},
  Ticks → {{{vislow, "V"}, {vishigh, "R"}, {0.968 * 10^-6, "0.97"},
      {2 * 10^-6, "2 μm"}, {3 * 10^-6, "3 μm"}, {4 * 10^-6, "4 μm"}, {5 * 10^-6, "5 μm"}},
    {{fvlow, "0.22T"}, {fvhigh, "2.23T"}, {3.105 * 10^12, "3.11T"}}},
  PlotLabel → "3000K Blackbody Radiation Curve With Visible Band"]
```

Out[17]=



Plot it again, zooming in on the inner micrometer so the visible band and peak are more obvious.

In[18]:=
```
PPlankZoom = Plot[fluxdensity[λ], {λ, 100 * 10^-9, 1050 * 10^-9},
  BaseStyle → {FontColor → RGBColor[0, 0.5, 0], FontSize → 12},
  PlotRange → {Automatic, 4 * 10^12}, PlotStyle → {Brown},
  AxesLabel → {"Wavelength μm", "Intensity W/m^2/λ"},
  GridLines -> {{400 * 10^-9, 700 * 10^-9, 0.96765 * 10^-6}, {2.23 * 10^11, 2.33 * 10^12, 3.105 * 10^12}},
  Ticks → {{{400 * 10^-9, "V"}, {700 * 10^-9, "R"}, {0.968 * 10^-6, "0.97 μm"}},
    {{2.23 * 10^11, "0.22T"}, {1 * 10^12, "1x10^12"}, {2 * 10^12, "2x10^12"},
     {2.33 * 10^12, "2.23T"}, {3.105 * 10^12, "3.11T"}, {4 * 10^12, "4x10^12"}}},
  PlotLabel → "3000K Blackbody Radiation Curve With Visible Band"]
```

Out[18]=



Create plots that will shade the UV, Visible and IR domains.

In[19]:=
```
PIR = Plot[fluxdensity[λ], {λ, 700 * 10^-9, 3 * 10^-6}, PlotRange → {Automatic, 4 * 10^12},
   PlotStyle → {Brown}, Filling → 0, FillingStyle → RGBColor[1.00, 0.82, 0.82]];
PVisible = Plot[fluxdensity[λ], {λ, 400 * 10^-9, 700 * 10^-9}, PlotRange → {Automatic, 4 * 10^12},
   PlotStyle → {Brown}, Filling → 0, FillingStyle → RGBColor[0.75, 1.00, 0.75]];
PUV = Plot[fluxdensity[λ], {λ, 10 * 10^-9, 400 * 10^-9}, PlotRange → {Automatic, 4 * 10^12},
   PlotStyle → {Brown}, Filling → 0, FillingStyle → RGBColor[0.6, 0.40, 1]];
```

In[22]:=
```
Show[PPlanck, PIR, PVisible, PUV,
 PlotLabel → "3000K Blackbody Radiation Curve With UV, Visible and IR Bands"]
```

Out[22]=



a) Now find the power emitted in the visible by integrating under the curve between V (the smallest wavelength in the violet) and R (the largest wavelenght in the red).

In[23]:=
```
IVIS = N[Integrate[fluxdensity[λ], {λ, 400 * 10^-9, 700 * 10^-9}]]
```

Out[23]= 367 074.

Find the power emitted in the UV by integrating under the curve between 10 and 400 nm.

In[24]:=
```
IUV = N[Integrate[fluxdensity[λ], {λ, 10 * 10^-9, 400 * 10^-9}]]
```

Out[24]= 9595.78

Find the power emitted in the IR by integrating under the curve between 700 nm and 3000 nm (3 μm).

In[25]:=
```
IIR = N[Integrate[fluxdensity[λ], {λ, 700 * 10^-9, 3 * 10^-6}]]
```

Out[25]= 3.68844×10^{6}

Find the total power emitted by integrating under the curve between 10 and 3000 nm.

In[26]:=
```
Itotal = N[Integrate[fluxdensity[λ], {λ, 10 * 10^-9, 3000 * 10^-9}]]
```

Out[26]= 4.06511×10^{6}

Compare the Power output in each band.

In[27]:=
```
VisibleToIR = IVIS / IIR
VisibleToUV = IVIS / IUV
VisibleToTotal = IVIS / Itotal
```

Out[27]= 0.0995201

Out[28]= 38.2537

Out[29]= 0.0902987

So the visible light is only about 9% of the total energy emitted by an incandescent bulb at 3,000 K. It's about 10% of the energy emitted in IR, so there really is more heat than light!! The visible is 38 times greater than the UV emission ... good thing, we'd all be blind!

b) Now compare the intensity emitted at 400 nm, 700 nm and 970 nm (0.970 μm).

In[30]:=
```
VtoPeak = fvlow / Imax
RtoPeak = fvhigh / Imax
```

Out[30]= 0.0716565

Out[31]= 0.751717

Thus the violet intensity is only 7% of the peak intensity, whereas the red intensity is 75% of it. This is what gives incandescent light its yellowish glow ... much more red light emitted than blue.

In[32]:=
```
Export["TRex_Pr03-24.pdf", SelectedNotebook[]]
```

TRex_Pr03-24.pdf